Paper 253-27

Individual Growth Analysis Using PROC MIXED

Maribeth Johnson, Medical College of Georgia, Augusta, GA

ABSTRACT

Individual growth models are designed for exploring longitudinal data on individuals over time. PROC MIXED allows the growth parameters for each individual to be examined as random effects in the model. Individual-level covariates can be entered into the model as fixed effects to determine their impact on the dependent variable alone and in interaction with the growth parameters. The structure of the variance-covariance matrix of the repeated measurements can also be examined and entered into the model. A model building exercise will be demonstrated using up to eight systolic blood pressure measurements of youths aged 7-22.

INTRODUCTION

Essential hypertension (EH) is a major risk factor for coronary heart disease, which remains the leading cause of death in the USA. Prior to middle age, the prevalence of EH in African Americans (AAs) is approximately twice as in European Americans (EAs), and EH-related mortality rates are 10 times higher in AAs compared to EAs.

The early natural progression of the development of high systolic blood pressure (SBP) from childhood to adulthood within the context of ethnicity and sex is not completely understood. A better understanding might contribute to a better identification of people at risk and, in turn, lead to improved prevention of hypertension.

This study evaluated the effect of sex and ethnicity upon the development of SBP over an eight year period. The ages of the children over this eight years ranged from 7 to 22 years. The data consisted of 3187 records on 524 individuals (129 EA males, 128 EA females, 122 AA males, 273 AA females). To be included in the analysis an individual had to have at least 4 out of 8 possible SBP measurements denoted as WAVEs which are measurements taken a year apart. The frequency distribution of the number of measurements is seen in Table 1. Over half of the individuals had 7 or 8 measurements and this was consistent within the ethnicity-sex subgroups.

Table 1. Measurement count distribution

Number of SBP measurements	Frequency	Percent
4	156	29.8
5	37	7.1
6	65	12.4
7	140	26.7
8	126	24.1

Plots of mean SBP across AGE for the four ethnicity-sex subgroups can be seen in the Appendix along with the AGE by WAVE frequency table. WAVE is positively correlated with AGE but they are not mutually exclusive.

THE MODEL

Individual growth models are used for exploring longitudinal data on individuals over time. Random effects or multilevel modeling allows for investigation of two levels of variability of the response variable, SBP: within and between subjects. In longitudinal data, observations taken over time are nested within subjects drawn from some population of interest giving a two-level hierarchical structure. The variation of responses within subjects over time is at the lowest level (level one) and the variation of the underlying mean responses between subjects is at level two (Singer, 1998). Measurements made on the same individual are correlated and it is this dependency that leads to the inadequacy of simple estimation procedures based on ordinary least squares. Sometimes in longitudinal data the interest lies as much in the covariance matrix estimates as in the average growth parameters. Using this technique allows us to examine both.

UNCONDITIONAL GROWTH

Unconditional growth for SBP was modeled as a function of age as a deviation from 15. Expressing age as a deviation from its mean reduces the multicollinearity between linear and quadratic regression coefficients if a quadratic model is shown to best fit the data. The intercept and slopes are fit as random effects which vary across subjects.

Linear Growth

The following SAS® code is used to fit an unconditional linear growth model:

```
proc mixed noclprint covtest;
  class id;
  model sbp = age15 / solution ddfm=bw;
  random intercept age15 / sub=id type=un
gcorr;
run; quit;
```

The NOCLPRINT option on the PROC MIXED statement prevents the printing of the CLASS level information giving the number of children involved in the analysis. THE COVTEST option tells SAS that you would like hypothesis tests for the variance and covariance components.

The MODEL statement is used to indicate the fixed effects and the RANDOM statement is used to indicate the random effects. The /SOLUTION option asks SAS to print the estimates for the fixed effects. The random effects, here the intercept and the linear slope, are estimated within each subject and the variance components estimate the variation of the underlying mean responses between subjects (SUB=ID). The structure of this 2X2 variance-covariance (V-C) matrix is unstructured (UN) and the GCORR option tells SAS to print the estimated correlation matrix amongst the random effects.

Quadratic Growth

Since the plots of mean SBP across AGE appeared to be curvilinear a quadratic model was applied to the data. The code is as follows:

```
proc mixed noclprint covtest;
  class id;
  model sbp = age15 age15*age15 / solution
ddfm=bw;
  random intercept age15 age15*age15/sub=id
type=un gcorr;
run; quit;
```

The only addition is that of the quadratic age term both in the model and as a random effect. The dimension of the unstructured V-C matrix is now 3X3.

Quadratic growth: V-C grouped by ethnic-sex subgroup

The pattern across age for SBP appears quite different for the ethnicity-sex subgroups (see plots and tables in Appendix). The AA males show more increase in SBP and much more variation in their SBP over time. The V-C matrix of the random effects for each individual quadratic growth curve was grouped by ethnicity and sex to see if these differences are significant. The similarity of the V-C matrix among subgroups was examined. Differences in the variance between members of the same subgroup might indicate sampling problems or an inherent difference between groups.

The SAS code that groups the V-C matrix of the random effects of the intercept and slopes follows:

```
proc mixed noclprint covtest;
  class id race sex;
  model sbp = age15 age15*age15 / solution
ddfm=bw;
  random intercept age15 age15*age15/sub=id
type=un gcorr group=race*sex;
run; quit;
```

RACE and SEX must be added to the CLASS statement and the GROUP=RACE*SEX option is added to the RANDOM statement.

Unconditional Growth Model Comparisons

A likelihood ratio test (LRT) for the significance of a more general model can be constructed if one model is a submodel of another by computing -2 times the difference between their residual log likelihoods (-2RLL). Then this statistic is compared to the chi-square distribution with degrees of freedom equal to the difference in the number of parameters for the two models. Models are preferred where the -2RLL is smaller. Model comparisons can also be made using Akaike's Information Criterion (AIC) or Schwarz's Bayesian Criterion (SBC). In both cases, the model that has the largest value is the preferred model. SBC penalizes models with more covariance parameters more than AIC so the two criteria may not agree when assessing model fit.

Table 2 shows the estimates for the fixed effects of these three models and Table 3 shows the comparisons between models for unconditional growth. A model with a random linear and quadratic effect provided a significantly better fit than the model with the random linear component alone based on all three model comparison methods. Adding a cubic component did not significantly improve the fit of the model (data not shown).

Table 2. Unconditional Growth Model: Fixed Effect Estimates (SE)

Model	Intercept	Age15	$Age15^2$
Linear	109.9 (.35)	.92 (.06)	---
Quadratic	110.3 (.37)	.85 (.06)	-.055 (.01)
Quadratic: V-C grouped	109.8 (.36)	.80 (.06)	-.058 (.01)

The quadratic model that grouped the V-C matrix on ethnic-sex subgroup provided a significantly better fit over the model where the where the V-C matrix was ungrouped (Table 3) although the estimates of the fixed effects were similar for the two models (Table 2).

Table 3. Unconditional Growth Model: Model Comparisons

Model	Param	AIC	SBC	-2RLL	Chi-square / df
Linear	3	-11014	-11022	22020	----
Quadratic	6	-10998	-11013	21982	38 / 3 *
Quadratic: V-C grouped	24	-10994	-11048	21940	42/18 *

*p<.001

The variances for both quadratic models are shown in Table 4. When broken out into groups it can be seen that the variation in SBP growth between AA males, both in the average SBP at age 15 and in the linear rate of growth, is much larger than the other subgroups. The between subject variation for EA males and females and for AA females appear similar. The covariances between the random effects are not shown but there is a significant positive correlation between the intercept and the linear slope showing that those with higher average SPB at age 15 are also increasing at a faster rate across age. The correlation between the linear and quadratic slope coefficients is 0.

There is variation in both the intercepts and the linear slopes that potentially could be explained by a level 2 (subject-level) covariate and this variation is different within ethnicity-sex subgroups. The unconditional growth model where the V-C matrix of the random effects is grouped by ethnicity-sex is used to investigate the effects of covariates on SBP changes over time.

Table 4. Unconditional Quadratic Growth Model: Variance (SE) of the Random Effects

Model	Subgroup	Intrcpt	Age15	$Age15^2$	Resid
V-C ungrouped		60.4 (4.5)	.47 (.12)	.010 (.004)	37.7 (1.2)
V-C grouped	EA male	47.8 (7.4)	.40 (.21)	.005 (.007)	37.6 (1.1)
	EA female	56.2 (9.0)	.37 (.23)	.001 (.008)	
	AA male	100.6 (15.2)	.79 (.29)	.020 (.011)	
	AA female	43.1 (6.3)	.30 (.22)	.020 (.010)	

ADDITION OF SUBJECT-LEVEL COVARIATES

Models were considered that investigated the fixed effect of different covariates on SBP as well as whether the variation in intercepts and slopes of the individual growth curves was related to these covariates. When investigating different covariates it is best not to group the V-C matrix due to computational and time constraints. Once a covariate is shown to have a significant effect then the V-C matrix can be grouped to show which subgroup, if any, was most effected by the addition.

Ethnicity

The code used to fit ethnicity in the model follows (AA is the reference group in the solution vector):

```
proc mixed noclprint covtest;
  class id race sex;
  model sbp = age15 age15*age15 race
race*age15 race*age15*age15 / solution
ddfm=bw;
  random intercept age15 age15*age15/sub=id
type=un gcorr group=race*sex;
  lsmeans race / pdiff;
run; quit;
```

The results are presented in Table 5. The interaction between ethnicity and the quadratic term is not included since it was not a significant effect.

Table 5. Addition of Ethnicity to the Quadratic Growth Model for SBP

	Ethnicity: AA is reference group
Fixed Effect Estimates (SE)	
Intercept	111.2 (.49)

Age15	.95 (.08)
$Age15^2$	-.055 (.012)
Covariate	-3.1 (.66)
Covariate*Age15	-.34 (.11)
Random Effect Estimates (SE)	
Intercept : EA male	56.3 (9.2)
EA female	44.9 (7.6)
AA male	88.2 (13.6)
AA female	45.6 (6.9)
Age15: EA male	.51 (.24)
EA female	.21 (.21)
AA male	.65 (.28)
AA female	.46 (.25)

There is a significant mean SBP difference due to ethnicity. The average SBP at age 15 is 108.1 for EA and is 111.2 for AA. There is also a significant interaction between ethnicity and the linear age coefficient where the rate of increase in SBP across age is .61mmHg/year for EA and .95mmHg/year for AA. The variation in the intercepts and slopes was slightly reduced for EA females (15%) and AA males (12%), slightly increased for EA males (18%), and unchanged for AA females. Adding ethnicity to the quadratic growth model helped the estimation of the curves for EA females and AA males.

Sex

The code used to fit sex in the model follows (Female is the reference group in the solution vector):

```
proc mixed noclprint covtest;
  class id race sex;
  model sbp = age15 age15*age15 sex sex*age15
sex*age15*age15 / solution ddfm=bw;
  random intercept age15 age15*age15/sub=id
type=un gcorr group=race*sex;
  lsmeans sex / pdiff;
run; quit;
```

The results are presented in Table 6. The interaction between sex and the quadratic term is not included since it was not a significant effect.

Table 6. Addition of Sex to the Quadratic Growth Model for SBP

	Sex: Female is reference group
Fixed Effects Estimates (SE)	
Intercept	107.6 (.44)
Age15	.38 (.08)
$Age15^2$	-.053 (.011)
Covariate	5.1 (.64)
Covariate*Age15	.90 (.10)
Random Effects Estimates (SE)	
Intercept : EA male	46.5 (7.3)
EA female	42.6 (6.9)
AA male	79.0 (12.0)
AA female	47.2 (7.0)
Age15: EA male	.36 (.21)
EA female	.09 (.18)
AA male	.46 (.23)
AA female	.18 (.21)

There is significance due to the main effect of sex where the average SBP at age 15 for males is 112.7 and for females is 107.6. There is also a significant difference in the rate of SBP change across age where males increase at 1.28mmHg/year and females at .38mmHg/year. The addition of sex to the model resulted in a reduction in the intercept variance for EA females (24%) and AA males (20%) while this variance was unchanged for EA males and AA females. The variation in slopes was greatly reduced between EA females (76%), AA males (42%) and AA females (40%) and relatively unchanged between EA males. The addition of sex to the quadratic growth model helped the estimation of the curves for all subgroups except EA males.

Maximum Household Education

The maximum education level attained by either the mother or the father of the child was used to try to assess socioeconomic status. This variable was coded as LOW for those whose parents had a high school or less education and HIGH otherwise.

The code used to fit education level in the model follows (Low parental education is the reference group in the solution vector):

```
proc mixed noclprint covtest;
  class id race sex ed;
  model sbp = age15 age15*age15 ed ed*age15
ed*age15*age15 / solution ddfm=bw;
  random intercept age15 age15*age15/sub=id
type=un gcorr group=race*sex;
  lsmeans ed / pdiff;
run; quit;
```

The results are presented in Table 7. The interaction between education and the quadratic term is not included since it was not a significant effect.

Table 7. Addition of Maximum Household Education to the Quadratic Growth Model for SBP

	Education: LOW is reference group
Fixed Effects Estimates (SE)	
Intercept	110.9 (.62)
Age15	1.08 (.10)
$Age15^2$	-.056 (.011)
Covariate	-1.5 (.72)
Covariate*Age15	-.40 (.12)
Random Effects Estimates (SE)	
Intercept : EA male	47.4 (7.4)
EA female	55.9 (9.0)
AA male	98.7 (15.0)
AA female	43.1 (6.4)
Age15: EA male	.39 (.21)
EA female	.39 (.22)
AA male	.69 (.28)
AA female	.31 (.22)

Those subjects whose parents had achieved a level of education beyond high school had a slightly lower average SBP at age 15 (109.4) and lower rate of SBP increase across age (.68mmHg/year) than those whose parents had a lower level of education (110.9 and 1.08mmHg/year, respectively). The variation in intercepts was unchanged for all groups by the addition of maximum household education. The linear slope variation was unchanged for EA and AA females but was slightly reduced, and therefore aided the slope estimation, for EA (17%) and AA (13%) males.

Full Model

We then looked at the effect of ethnicity, sex and education together on SBP growth. Height in centimeters centered at the mean of 150 (HTCM150) and BMI centered at its mean of 30 (BMI30) were included in the model to control for differences in growth and adiposity. It was thought that some of the ethnicity and sex differences might be due to these factors. The parameter estimates and the random effect variances are shown in Table 8.

Table 8. Full model

	Fixed Effect Estimates (SE)	Type III test p-value
Intercept	109.3 (.81)	-
Age15	.29 (.13)	.37
$Age15^2$	.034 (.016)	.03
Ethnicity *	-3.2 (.60)	<.0001
Sex **	2.4 (.55)	<.0001
Education ***	-1.1 (.64)	.10
Ethnicity*Age15	-.41 (.10)	<.0001
Education*Age15	-.35 (.11)	.0013
HTCM150	.16 (.03)	<.0001
Sex*HTCM150	.14 (.02)	<.0001
BMI30	.30 (.04)	<.0001
	Random Effects Estimates (SE)	
Intercept: EA male	33.6 (5.5)	
EA female	33.5 (5.5)	
AA male	65.2 (9.9)	
AA female	44.6 (6.6)	
Age15: EA male	.16 (.18)	
EA female	.10 (.17)	
AA male	.55 (.25)	
AA female	.22 (.20)	

*AA is the reference group
**Female is the reference group
***Low parental education is the reference group

HTCM150 and BMI30 are both time dependent covariates that are highly correlated with age. The linear effect of age is no longer significant in this model since HTCM150 and BMI30 have entered the model as significant effects. There is still a significant quadratic age effect. EA have significantly lower average SBP at age 15 (106.1) and males have higher average SBP at age 15 (111.7) than AA or females (109.3). There was not a significant interaction between ethnicity and sex for means or linear slopes. The rate of increase across age in SBP, adjusted for differences in growth and adiposity, was significantly less for EA and for those whose parents achieved a level of education beyond high school. As all subjects grew taller their SBP increased, although males increased at a significantly faster rate than females. Increases in body mass were also related to increases in SBP for all subjects.
This model provides a 30-40% reduction in the variation of the intercepts (average SBP at age 15) for EA males and females and AA males. The variance of the intercepts remained unchanged from the unconditional growth model for AA females.

The between subject variance in the linear slopes has been reduced from 30-70% by this model. The estimate of the variance does not exceed its standard error for EA males and females and AA females. There is still significant variation in the slope estimates for AA males that is not explained by this model.

WITHIN-SUBJECT COVARIANCE STRUCTURE

One of the strengths of PROC MIXED is that it allows the comparison of different structures for the within subject, or error, covariance matrix. So far we have been concerned with the between-subject variances of the intercepts and slopes of the growth model but have not been concerned about the within-subject variances for the up to eight SBP measurements (WAVEs). In this context, the intercepts and growth rates are assumed to be constant across individuals but the model introduces a different type of complexity: the residual observations within subjects (after controlling for the linear and quadratic effects of AGE) are correlated through the within-subject error V-C matrix. This error matrix is called the **R** matrix and is modeled using the REPEATED statement.

```
proc mixed noclprint covtest;
  class id wave;
  model sbp = age15 age15*age15 / s notest;
  repeated wave /sub=id r rcorr type=cov-
structure;
 run; quit;
```

WAVE has been added as a CLASS variable to indicate the time structured nature of the data within subjects and WAVE is also used on the REPEATED statement. The SUB=ID option is the mechanism for block diagonalizing **R** since subjects are considered independent. The R option of the REPEATED statement requests that the first block of the **R** matrix be printed, the RCORR option prints the correlation matrix corresponding to **R**. If the first subject does not have all of the measurements (WAVEs) then you will need to specify a subject who does.

The TYPE= option is what determines the V-C structure. Five different *cov-structures* were considered, these are:

1) Compound symmetric (CS): This is the most specific structure, the variance within waves is constant and there is a common correlation between waves. This is the assumption if only the intercepts vary across individuals. There are two parameters estimated.
2) Heterogeneous compound symmetric (CSH): This structure assumes a common correlation between waves but allows for different variances along the diagonal. For these data there are 9 parameters estimated.
3) Toeplitz (TOEP): This structure assumes a common variance across waves but produces a banded covariance structure such that the correlations between waves separated by the same amount of time are equal. There are 8 parameters estimated.
4) Heterogeneous Toeplitz (TOEPH): This structure allows for different variance parameters and produces a banded covariance structure such that the correlations between waves separated by the same amount of time are equal. There are 15 parameters estimated.
5) Unstructured (UN): This structure produces estimates of all eight variances and 28 covariances in each subject block of **R**. Therefore, all of the correlations between waves may be different.

Table 9 shows the comparisons between these models. Since the same fixed effects were included in all models a likelihood ratio test (LRT) was used to compare models for which one is a special case of the other. CSH and TOEP were compared to CS, TOEPH was compared to TOEP, and UN was compared to TOEPH.

Table 9. Within-Subject Model Comparisons

Type	Parameters	AIC	SBC	-2RLL	Chi-square / df
CS	2	-11053	-11057	22102	---
CSH	9	-11042	-11061	22066	36 / 7 *
TOEP	8	-11022	-11039	22028	74 / 6 *
TOEPH	15	-11015	-11045	22000	28 / 7 *
TOEPH: grouped	60	-11001	-11128	21881	119/45 *
UN	36	-11015	-11092	21959	41 / 21

*p<.001

Recall that we prefer models in which the AIC and SBC are larger and the -2RLL is smaller. Allowing for heterogeneous variances for each wave along the diagonal provides a significantly better fit for both the CS and the TOEP models. Allowing for a banded covariance structure (TOEP) provides a better fit than the assumption that all of the correlations between waves are equal (CS). The LRT between UN and TOEPH did not show a significant difference in fit so the preferred model for these data is one where the error matrix has a TOEPH structure.

Since the variances of the intercepts and the slopes were significantly different within the AA male subgroup, a model was constructed that looked into possible subgroup differences in the **R** matrix using the TOEPH structure.

```
proc mixed noclprint covtest;
  class id wave race sex;
  model sbp = age15 age15*age15 / s notest;
  repeated wave /sub=id r rcorr toeph
group=race*sex;
 run; quit;
```

The model comparison results of the TOEPH model grouped by ethnicity and sex are also shown in Table 9. The grouping provides a significantly better fit over not grouping, even though there are 45 more parameters to estimate.

The variance and correlation estimates for each subgroup are listed in Table 10. The TOEPH(X) values are the correlation between values that are X waves apart, so TOEPH(1) is the estimated correlation between waves 1 and 2, 2 and 3, 3 and 4, 4 and 5, 5 and 6, 6 and 7, and 7 and 8.

Table 10. V-C Parameter estimates (SE) from the TOEPH model: Grouped by Ethnicity and Sex

	EA Male	EA Female	AA Male	AA Female
Var (1)	58 (9)	75 (12)	120 (18)	103 (15)
Var (2)	88 (13)	89 (14)	128 (18)	83 (12)
Var (3)	79 (12)	82 (12)	130 (18)	55 (7)
Var (4)	59 (8)	68 (9)	147 (21)	72 (9)
Var (5)	80 (10)	83 (11)	161 (21)	86 (10)
Var (6)	86 (11)	70 (9)	172 (23)	84 (10)
Var (7)	94 (12)	75 (10)	170 (23)	101 (12)
Var (8)	93 (16)	92 (16)	170 (27)	139 (21)
TOEPH(1)	.59 (.04)	.66 (.04)	.72 (.03)	.49 (.04)
TOEPH(2)	.54 (.04)	.63 (.04)	.68 (.04)	.48 (.04)
TOEPH(3)	.51 (.05)	.59 (.04)	.67 (.04)	.46 (.04)
TOEPH(4)	.47 (.06)	.59 (.05)	.62 (.05)	.43 (.05)
TOEPH(5)	.42 (.07)	.46 (.07)	.64 (.05)	.38 (.06)
TOEPH(6)	.38 (.08)	.48 (.07)	.62 (.06)	.39 (.07)
TOEPH(7)	.48 (.10)	.52 (.10)	.71 (.07)	.55 (.08)

The AA Male group also shows much larger within WAVE variation in SBP than do the other groups. Their values also seem to track slightly better across WAVE but these correlations are only marginally higher when compared to the other the other groups. The major difference in subgroups appears to be within-WAVE variation in SBP.

COMBINED MODEL

When the within-subject error covariance matrix specification was combined with the model where the intercepts and slopes were considered as random effects the new model did not converge. This occurred even when neither V-C matrix was grouped by ethnicity and sex.

This new model is probably over specified. The error covariance structure within subjects describes the behavior of the errors - what remains after removing other fixed and random effects in the model. The specification of the TOEPH error matrix may no longer be needed after the intercepts and slopes are considered as random effects and additional covariates are added to the model.

CONCLUSION

PROC MIXED allows for the investigation of the changes over time within and between individuals. The intercepts and slopes of the individual growth curves can be considered as random effects and the effects of adding covariates to the model can be evaluated as changes in variances as well as the fixed effects estimates of those growth parameters and covariates.

Many different models can be considered and comparisons made in order to try to determine the best fit for the data. A limitation of this analysis technique for these data is that no one subject has measurements across the entire AGE spectrum. The change in SBP across AGE is derived from up to 8-year accumulations from many individuals.

REFERENCES

Littel, RC, Milliken, GA, Stroup, WW and Wolfinger, RD (1996) *SAS System for Mixed Models*, Cary, NC: SAS Institute, Inc.

SAS Institute Inc. (1989), *SAS/STAT User's Guide: Version 6, Fourth Edition, Volume 2,* Cary, NC: SAS Institute Inc.

SAS Institute Inc. (1992), *SAS Technical Report P-229, SAS/STAT Software: Changes and Enhancements, Release 6.07*, Cary, NC: SAS Institute Inc.

SAS Institute Inc. (1994), *SAS/STAT Software: Changes and Enhancements, Release 6.10*, Cary, NC: SAS Institute Inc.

Singer, JD (1998), *Using SAS PROC MIXED to Fit Multilevel Models, Hierarchical Models, and Individual Growth Models*, J Educational and Behavioral Statistics, 24(4): 323-355. (Also found at http://gseweb.harvard.edu/~faculty/singer/)

Wolfinger, RD (1992), "A tutorial on mixed models", Cary, NC: SAS Institute, Inc.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Maribeth Johnson
Office of Biostatistics AE-3035
Medical College of Georgia
Augusta, GA 30912-4900
Work Phone: (706) 721-0813
Fax: (706) 721-6294
Email: maribeth@stat.mcg.edu

APPENDIX









Subgroup Means Across Age: SBP

	EA Male		EA Female		AA Male		AA Female	
AGE	N	Mean (SD)	N	Mean (SD)	N	Mean (SD)	N	Mean (SD)
7	12	102 (7)	12	100 (7)	0	---	0	---
8	25	103 (7)	31	103 (7)	18	103 (7)	23	103 (13)
9	23	102 (6)	22	103 (7)	23	105 (8)	30	103 (8)
10	45	103 (7)	37	102 (7)	37	104 (7)	39	104 (9)
11	52	106 (9)	53	103 (7)	45	105 (8)	50	107 (8)
12	64	105 (7)	65	105 (8)	56	108 (9)	67	108 (8)
13	78	109 (9)	83	106 (7)	67	111 (10)	84	107 (8)
14	97	112 (9)	93	104 (8)	72	113 (10)	113	108 (9)
15	105	112 (9)	102	105 (9)	85	114 (10)	122	110 (9)
16	84	113 (9)	90	104 (8)	87	116(10)	111	111(9)
17	65	113 (8)	72	106 (9)	87	119 (11)	84	112 (12)
18	49	115 (10)	39	105 (8)	70	121 (13)	62	109 (9)
19	22	115 (11)	34	107(8)	46	122 (13)	42	111 (10)
20	22	114 (9)	24	108 (9)	32	121 (12)	31	111 (9)
21	16	115 (11)	12	109 (14)	19	122 (13)	21	116 (10)
22	14	117 (13)	0	---	11	127 (14)	11	113 (15)

The FREQ Procedure

Table of AGE by WAVE

AGE WAVE

Frequency	1	2	3	4	5	6	7	8	Total
7	23	1	0	0	0	0	0	0	24
8	85	12	0	0	0	0	0	0	97
9	53	36	8	1	0	0	0	0	98
10	44	75	35	3	1	0	0	0	158
11	36	78	60	21	4	1	0	0	200
12	28	37	83	65	33	4	1	1	252
13	25	26	37	109	76	35	3	1	312
14	25	25	24	67	121	76	33	4	375
15	17	23	35	49	76	116	76	22	414
16	11	15	21	34	63	80	113	35	372
17	3	11	23	38	45	64	81	43	308
18	0	4	16	18	39	40	63	40	220
19	0	0	3	13	17	41	41	29	144
20	0	0	1	4	15	23	36	30	109
21	0	0	0	2	3	12	23	28	68
22	0	0	0	1	2	3	12	18	36
Total	350	343	346	425	495	495	482	251	3187